

Hewlett-Packard

HP-19C/HP-29C SOLUTIONS

FINANCE

INTRODUCTION

This HP-19C/HP-29C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become an expert on your HP calculator.

You will find general information on how to key in and run programs under "A Word about Program Usage" in the Applications book you received with your calculator.

We hope that this Solutions book will be a valuable tool in your work and would appreciate your comments about it.

TABLE OF CONTENTS

DIRECT REDUCTION LOAN AMORTIZATION SCHEDULE

Loan Amt: \$45,000 APR: 9.25%
Monthly Payment: \$385

Payment Number	Paid to Interest	Paid to Principal	Remaining Balance
13	343.20	41.80	44480.80
14	342.87	42.13	44438.67
15	342.55	42.45	44396.22
18	341.56	43.44	44266.89
13 - 18	2054.29	255.71	44266.89

Given the periodic interest rate (i), periodic payment amount (PMT), loan amount (PV) and beginning and ending periods (P_1, P_2), this program will generate the values for a loan amortization schedule as pictured above, starting with the payment P_1 and ending with P_2. After P_2 has been reached the program calculates the accumulated interest and principle for the payments made between P_1-P_2 inclusively. The schedule may be started at any point in the mortgages life.

The data generated is valid for loans that have a balloon payment as well as those that are arranged to be fully amortized.

For loans with a balloon payment, the remaining balance of the last payment period is the balloon payment due in addition to the last periodic payment.

For loans scheduled to be fully amortized, the remaining balance after the last payment period may be slightly more or less than zero. This is because the program assumes that <u>all</u> payments are equal to the value entered for PMT. In fact for most loans the last payment is slightly more or less than the rest.

<u>Equations</u>:

$$n = -\frac{\ln\left(-\frac{iPV}{PMT} + 1\right)}{\ln(1 + i)}$$

$$BAL_{P_1-1} = PMT\left[\frac{1 - (1 + i)^{P_1-1-N}}{i}\right]$$

$$INT_{Pn} = RND\left(BAL_{Pn} \times \frac{i}{100}\right)$$

$$PRIN_{Pn} = PMT - INT_{Pn}$$

$$BAL_{Pn} = BAL_{Pn-1} - PRIN_{Pn}$$

Example:

Duplicate the entries in the preceding amortization schedule.

Solution:

45000.00	STO2	mortgage amount
9.25	ENT↑	annual percentage rate
12.00	÷	payments per year
	STO3	periodic interest rate
385.00	STO4	periodic payment amount
13.00	STO0	P_1
18.00	STO1	P_2
	GSB0	
301.14	***	actual life of mortgage
44522.60	***	remaining bal. at P_1-1*
13.00	***	payment number
343.20	***	interest portion of pmt.
41.80	***	principle portion of pmt.
44480.80	***	remaining after 13th pmt.
14.00	***	
342.87	***	
42.13	***	
44438.67	***	
15.00	***	
342.55	***	
42.45	***	
44396.22	***	
16.00	***	
342.22	***	
42.78	***	
44353.44	***	
17.00	***	
341.89	***	
43.11	***	
44310.33	***	
18.00	***	
341.55	***	
43.44	***	
44266.89	***	
2054.29	***	acc. interest periods 13-18
255.71	***	acc. principle periods 13-18
44266.89	***	remaining balance

*When using the HP-29C all subsequent values are produced by pressing R/S.

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in program				
2.	Enter following data in any order				
	- Original amount borrowed	PV	STO	2	
	- Periodic interest rate	i	STO	3	
	- Periodic payment amount	PMT	STO	4	
3.	Key in				
	- Starting period	P1	STO	0	
	- Ending period	P2	STO	1	
4.	Calculate life of mortgage		GSB	0	N
5.	Calculate remaining balance at P1		R/S*		Rem. Bal.
	- Period number		R/S*		Pn N
	- Principal payment		R/S*		Pn PRIN
	- Interest payment		R/S*		Pn INT
	- Remaining balance		R/S*		Pn BAL
6.	Repeat step 5 for each period				
7.	After last period (P2)				
	Calculate accumulated totals:				
	- Accumulated interest		R/S*		Σ INT
	- Accumulated principal		R/S*		Σ PRIN
	- Remaining balance		R/S*		Rem. Bal.
8.	For a new payment period in same mortgage				
	repeat steps 3-7.				
*	If the program is run on the HP-19C, the				
	R/S is not used.				

Program Listings

```
01 *LBL0
02    0
03  STO5
04  STO6
05  RCL3
06   EEX
07    2
08    ÷
09  STO7
10  RCL2
11    x
12  RCL4
13    ÷
14   CHS
15    1
16    +
17    LN
18  RCL7
19    1
20    +
21    LN
22    ÷
23   CHS   **
24   SPC   **
25  PRTX          "New" N
26  RCL0
27    1
28    -
29   X⇄Y
30    -
31  RCL7
32    1
33    +
34   X⇄Y
35    Yˣ
36    1
37   X⇄Y
38    -
39  RCL7
40    ÷
41  RCL4
42    x
43  STO9
44  PRTX   **     BAL P1 - 1
45   SPC   **
46 *LBL1
47  RCL1
48  RCL0
49  X>Y?
50  GSB3
51  PRTX   **     Pn
52   ISZ
53  RCL9
54  RCL3
55    %
56   EEX
57    7
58    +
59   EEX
60    7
61    -
62  PRTX   **     Pn INT
63  ST+5
64    +
65  RCL4
66  LSTX
67    -
68  PRTX   **     Pn PRIN
69  ST+6
70  LSTX
71    +
72    -
73  PRTX   **     Pn Rem. Bal.
74  STO9
75   SPC   **
76  GTO1
77 *LBL3
78  RCL5
79  PRTX   **     Σ INT
80  RCL6
81  PRTX   **     Σ PRIN
82  RCL9
83  PRTX   **     Rem. BAL P2
84   R/S
```

$$n = \frac{\ln\left(-\frac{iPV}{PMT} + 1\right)}{\ln(1+i)}$$

$$PMT\left[\frac{1 - (1+i)^{P1-1-N}}{i}\right]$$

** When using the HP-29C R/S must be used in the place of PRTX and SPC should be deleted.

REGISTERS

0 P_1	1 P_2	2 PV	3 Periodic i	4 PMT	5 Σ INT
6 Σ PRIN	7 i/100	8	9 Rem. BAL P_n	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

INTERNAL RATE OF RETURN
UP TO 26 CASH FLOWS

The interest rate that equates the present value of all future cash flows with the original investment is known as the Internal Rate of Return (also called discounted rate of return or yield). Given the initial investment and up to 26 cash flows, this program calculates the periodic IRR.

When using this program, cash received is positive and cash outlays are negative. Zero should be keyed in for periods in which there are no cash flows.

The answer produced is the periodic rate of return. If the cash flow period is other than annual, the answer should be multiplied by the number of periods per year to determine the annual internal rate of return.

The program solves the following equation iteratively for IRR:

$$INV = \sum_{j=1}^{n} \frac{CF_j}{(1 + IRR)^j}$$

where: n= number of cash flows

CF_j= j^{th} cash flow

Note:

Problems involving a large number of cash flows can often result in long execution times.

Example:

A shopping center requires a $200,000 investment, and will be sold at the end of 3 years. If this investment results in the semi-annual net cash flows shown, what is the internal rate of return?

End of Six Month Period	Cash Flow
1	$-50,000
2	0
3	11,000
4	11,000
5	13,000
6	280,000 (includes net proceeds from sale)

Solution:

	GSB1	
-50000.00	R/S	
0.00	R/S	
11000.00	R/S	
	R/S	
13000.00	R/S	
280000.00	R/S	
-200000.00	GSB2	Cash paid out
4.23	***	Semi-annual IRR (%)
2.00	x	
8.46	***	Annual IRR

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Clear registers		f	REG	
3.	Initialize		GSB	1	4
4.	Beginning with the first period,		R/S		CF_i
	enter all cash flows in sequence:				
	inflows positive, outflows negative,				
	and 0 for no cash flows.				
5.	Enter initial investment using the	INV	GSB	2	IRR(%)
	cash flow sign convention.				
6.	For a new case, go to step 2				

Program Listings

Step	Key	Comments
01	*LBL1	
02	4	
03	STO0	
04	*LBL0	store cash flows
05	R/S	
06	STOi	
07	ISZ	
08	GTO0	
09	*LBL2	
10	ABS	
11	DSZ	
12	STO1	INV
13	RCL0	n + 3
14	STO3	
15	1	initial value for
16	ENT↑	1 + IRR
17	0	
18	STO2	clear f(i)
19	*LBL6	
20	RCL0	
21	3	
22	X=Y?	
23	GTO8	
24	R↓	
25	3	
26	-	i
27	RCLi	CFi
28	GSB7	
29	DSZ	
30	GTO6	
31	*LBL8	
32	R↓	
33	R↓	
34	RCL3	
35	STO0	
36	R↓	
37	RCL2	f(i)
38	RCL1	
39	-	
40	X⇄Y	
41	÷	
42	x	Δ
43	+	1 + IRR
44	LSTX	
45	ABS	|Δ|
46	EEX	
47	CHS	
48	6	10^{-6}
49	X≤Y?	
50	GTO5	
51	R↓	
52	R↓	1 + IRR
53	1	
54	-	
55	EEX	
56	2	
57	x	
58	R/S	***IRR(%)
59	*LBL5	
60	R↓	
61	R↓	1 + IRR′
62	0	
63	STO2	
64	GTO6	
65	*LBL7	CFi i~ 1 + IRR
66	ST+2	
67	x	
68	+	
69	X⇄Y	1 + IRR
70	ST÷2	
71	÷	~1 + IRR 1 + IRR
72	RTN	1 + IRR

*** "Print x" may be inserted.

REGISTERS

0 i	1 INV	2 f(i)	3 n + 3	4 CF_1	5 CF_2
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29 CF_{26}

STRAIGHT LINE DEPRECIATION SCHEDULE

Schedule - Straight-Line Method

Starting Book Value: $375,000 Salvage Value $30,000

Estimated Useful Life: 40.25 Years

Year (End of)	Depreciation Amount (DEP)	Remaining Depreciable Value (RDV)	Remaining Book Value (RBV)	Depreciation To Date (Reserve)
1	8571.43	336428.57	366428.57	8571.43
2	8571.43	327857.14	357857.14	17142.86
41	2142.96	0.00	30000.00	345000.00

The annual depreciation allowance using this method is determined by dividing the cost or other basis of valuation (starting book value) less its estimated salvage value by its useful life expectancy. This program develops the data shown in the example schedule, given the starting book value (SBV), salvage value (SAL), life expectancy (LIFE), and first year of the schedule (YR), (The schedule may be started at any point in the useful life.)

Fractional years lives must be entered as an integer plus a fraction. Thus a life of 12 years 3 months would be keyed in as 12.25 for LIFE.

Values for the last year of an asset with fractional years life (i.e., the 21st year's values for an asset with 20.5 years life) are calculated correctly. However, all other values represent a full year's depreciation.

For this reason only integer values (whole number, 1.0, 2.0, 17.0 etc.) may be entered for YR. The program makes no checks on this value and generates invalid results if other than whole numbers are entered.

EQUATIONS:

$$DEP_k = \frac{SBV - SAL}{LIFE}$$

$$DEP_k(\text{last year}) = \left(\frac{SBV - SAL}{LIFE}\right) \cdot F$$

$$RES = (K) \cdot \left(\frac{SBV - SAL}{LIFE}\right)$$

$$RDV_k = (LIFE-K) \cdot \left(\frac{SBV - SAL}{LIFE}\right)$$

$$RBV_k = RDV_k + SAL$$

where

RES = Reserve

F = Decimal portion of LIFE

K = Value for YR

EXAMPLE: Complete the schedule shown for the first two years. Then jump to the 41st year and generate the data for that year.

SOLUTION:

```
    40.25 ENT↑
 30000.00 ENT↑
375000.00 ENT↑
     1.00 GSB1
          R/S
  8571.43 ***    DEP1
          R/S
336428.57 ***    RDV1
          R/S
366428.57 ***    RBV1
          R/S
  8571.43 ***    RES1
          R/S
  8571.43 ***    DEP2
          R/S
327857.14 ***    RDV2
          R/S
357857.14 ***    RBV2
          R/S
 17142.86 ***    RES2
    41.00 STO0   YR
          R/S
  2142.86 ***    DEP41
          R/S
     0.00 ***    RDV41
          R/S
 30000.00 ***    RBV41=SAL
          R/S
345000.00 ***    RES41
          R/S
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter data				
2a.	Life of asset in years	LIFE	ENT ↑		
2b.	Salvage value	SAL	ENT ↑		
2c.	Starting book value (beginning basis)	SBV	ENT ↑		
2d.	Year for which depreciation is to be				
	calculated	YR	GSB	1	
3.	RUN:				
3a.	Depreciation		R/S		DEP
3b.	Remaining depreciable value		R/S		RDV
3c.	Remaining book value		R/S		RBV
3d.	Total depreciation to date (Reserve)		R/S		RES
4.	For next year's values, go to step 3a				
5.	For another year's values, enter year and				
	go to step 3a	YR	STO	0	

Program Listings

Step	Key	Remarks
01	*LBL1	
02	STO0	
03	R↓	
04	STO1	
05	R↓	
06	STO2	
07	R↓	
08	STO3	
09	R/S	
10	*LBL5	
11	1	
12	+	
13	X≤Y?	
14	GTO9	Beyond useful life
15	FRC	
16	STO4	
17	RCL3	
18	RCL0	
19	÷	
20	1	
21	X>Y?	Last year?
22	RCL4	
23	1	
24	×	1 or F
25	RCL3	
26	÷	
27	RCL1	
28	RCL2	
29	-	
30	STO5	
31	×	
32	R/S	*** DEP
33	RCL3	
34	RCL0	
35	X>Y?	
36	GTO8	Last year?
37	-	
38	RCL3	
39	÷	
40	RCL5	
41	×	
42	STO6	
43	R/S	*** RDV
44	*LBL7	
45	RCL6	
46	RCL2	
47	+	
48	R/S	*** RBV
49	RCL5	
50	RCL6	
51	-	
52	ISZ	
53	R/S	*** RES
54	RCL0	
55	RCL3	
56	GTO5	
57	*LBL9	
58	0	
59	÷	Error
60	*LBL8	
61	0	
62	STO6	
63	R/S	*** RDV = 0
64	GTO7	
65	R/S	

*** "Printx" may be inserted to replace "R/S".

REGISTERS

0 YR	1 SBV	2 SAL	3 LIFE	4 F	5 SBV-SAL
6 RDV	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

SUM OF THE YEARS' DIGITS

Depreciation Schedule - Sum of the Years Digits Method
Starting Book Value: $375,000 Salvage Value: $30,000
Expected Useful Life: 40.25 Years

Year (End of)	Depreciation Amount (DEP)	Remaining Depreciable Value (RDV)	Remaining Book Value (RBV)	Depreciation To Date (Reserve)
1	16,725.38	328,274.62	358,274.62	16,725.38
2	16,309.85	311,964.77	341,964.77	33,035.23
41	103.88	0.00	30,000.00	345,000.00

The sum-of-the-years' digits method is an accelerated form of depreciation, allowing more depreciation in the early years of an asset's life than allowed under the straight line method. This program generates the data shown in the example schedule, given the starting book value (SBV), the salvage value (SAL), expected useful life in years (LIFE), and beginning year (YR) for the schedule. (The schedule may be started at any point in the useful life.)

Fractional years asset life must be entered as an integer plus a fraction. Thus a life of 12 years 3 months would be keyed in as 12.25 for LIFE.

Values for the last year of an asset with fractional years life (i.e., the 21st year's values for an asset with 20.5 years life) are calculated correctly. However, all other values represent a full year's depreciation. For this reason only integer values (whole numbers, 1.0, 2.0, 17.0, etc.) may be entered for YR. The program makes no checks on this value and generates invalid results if other than whole numbers are entered.

EQUATIONS:

$$\mathrm{SOYD} = \frac{(W + 1)(W + 2F)}{2}$$

$$\mathrm{DEP}_k = \left(\frac{\mathrm{LIFE} + 1 - K}{\mathrm{SOYD}}\right) \cdot (\mathrm{SBV-SAL})$$

$$\mathrm{RES}_k = \left[1 - \frac{(W-K+1)x(W-K+2F)}{2x(\mathrm{SOYD})}\right] \cdot (\mathrm{SBV-SAL})$$

$$\mathrm{RDV}_k = \left[\frac{(W-K+1)x(W-K+2F)}{2x(\mathrm{SOYD})}\right] \cdot (\mathrm{SBV-SAL})$$

$$\mathrm{RBV}_k = \mathrm{RDV}_k + \mathrm{SAL}$$

Where

K = value for YR

RES_k = Reserve at period k

W = Integer portion of LIFE

F = Decimal portion of LIFE

EXAMPLE: Complete the schedule shown for the first two years. Then jump to the 41st year and generate the data for that year.

SOLUTION:

40.25	ENT↑	
30000.00	ENT↑	
375000.00	ENT↑	
1.00	GSB1	
	R/S	
16725.38	***	DEP_1
	R/S	
328274.62	***	RDV_1
	R/S	
358274.62	***	RBV_1
	R/S	
16725.38	***	RES_1
	R/S	
16309.85	***	DEP_2
	R/S	
311964.77	***	RDV_2
	R/S	
341964.77	***	RBV_2
	R/S	
33035.23	***	RES_2
41.00	STO0	
	R/S	
103.88	***	DEP_{41}
	R/S	
0.00	***	RDV_{41}
	R/S	
30000.00	***	RBV_{41}
	R/S	
345000.00	***	RES_{41}

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter data				
2a.	Life of asset in year	LIFE	ENT ↑		
2b.	Salvage value	SAL	ENT ↑		
2c.	Starting book value (beginning basis)	SBV	ENT ↑		
2d.	Year for which depreciation is to be				
	calculated	YR	GSB	1	
3.	RUN:				
3a.	Depreciation		R/S		DEP
3b.	Remaining depreciable value		R/S		RDV
3c.	Remaining book value		R/S		RBV
3d.	Total depreciation to date (Reserve)		R/S		RES
4.	For next year's values, go to step 3a				
5.	For another year's values, enter year and				
	go to step 3a	YR	STO	0	

Program Listings

01 *LBL1	
02 STO0	
03 R↓	
04 STO1	
05 X⇄Y	
06 -	
07 STO8	
08 R↓	
09 STO3	
10 GSB9	
11 STO7	SOYD
12 R/S	
13 *LBL5	
14 RCL0	
15 RCL3	
16 1	
17 +	
18 X≤Y?	
19 GTO8	Beyond useful life
20 X⇄Y	
21 -	
22 RCL7	
23 ÷	
24 RCL8	
25 x	
26 R/S	*** DEP
27 RCL3	
28 RCL0	
29 -	
30 X<0?	
31 GTO7	
32 GSB9	
33 RCL7	
34 ÷	
35 RCL8	
36 x	
37 STO5	
38 R/S	*** RDV
39 *LBL6	
40 RCL8	
41 RCL5	
42 -	
43 STO5	RES
44 RCL1	
45 X⇄Y	
46 -	
47 R/S	
48 RCL5	*** RBV
49 ISZ	
50 R/S	*** RES
51 GTO5	Next year or new year
52 *LBL9	
53 ENT↑	
54 FRC	
55 ENT↑	
56 +	2F
57 X⇄Y	
58 INT	
59 +	
60 LSTX	W or W-k
61 1	
62 +	
63 x	
64 2	
65 ÷	
66 RTN	
67 *LBL8	
68 0	
69 ÷	Error
70 *LBL7	
71 0	
72 STO5	
73 R/S	*** RDV = 0
74 GTO6	

*** "PRINTX" may be inserted to replace "R/S".

REGISTERS

0 YR	1 SBV	2	3 LIFE	4	5 RDV/RES
6	7 SOYD	8 SBV-SAL	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

VARIABLE RATE DECLINING BALANCE DEPRECIATION SCHEDULE

Depreciation Schedule - Declining Balance Method

Starting Book Value: $375,000 Salvage Value $30,000

Expected Useful Life: 40 Years Rate: 1.5

Year (End of)	Depreciation Amount (DEP)	Remaining Depreciable Value (RDV)	Remaining Book Value (RBV)	Depreciation To Date (Reserve)
1	14,062.50	330,937.50	360,937.50	14,062.50
2	13,535.16	317,402.34	347,402.34	27,597.66
15	8,235.18	181,369.51	211,369.51	163,630.49

The variable rate declining balance method is another form of accelerated depreciation; as such it provides for more depreciation in earlier years and decreasing depreciation in later years. This program generates the data shown in the example schedule given the starting book value (SBV), salvage value (SAL), useful life expectancy (LIFE), the declining rate factor (FACT), and the first year of the desired schedule (YR). The schedule may be started at any point in the useful life.

The "variable rate" is indicated as either a factor or percent with equal frequency in the business community. Thus, "1.5 declining balance factor" and "150% declining balance" have the same meaning. The number to be keyed in for FACT in this program, should be in factor form, that is 1.25, 1.5, 2, and not 125, 150 or 200.

This method of depreciation is unique in that it may generate depreciation greater than the depreciable value for some assets, while it may not generate sufficient depreciation for others.

This program will not allow an asset to be depreciated below its salvage value. That is when the generated depreciation for a year, usually the last, is greater than the remaining depreciable value, the latter is displayed as the depreciation amount. Program 6 is provided to assist in determining the best time to switch to straight line depreciation (tax laws permitting) so that an asset may be fully depreciated.

Fractional years lives must be entered as an integer plus a fraction however. Thus, a life of 12 years 3 months would be keyed in as 12.25.

Values for the last year of an asset with fractional years life (i.e., the 21st year's values of an asset with 20.5 years life) are calculated correctly. However, all other values represent a full year's depreciation. For this reason only integer values (whole numbers 1.0, 2.0, 17.0, etc.) may be entered for YR. The program makes no checks on this value and will generate invalid results if other than whole numbers are entered.

EQUATIONS:

$$DEP_k = SBV \cdot \left(1-\frac{FACT}{LIFE}\right)^{k-1} \cdot \left(\frac{FACT}{LIFE}\right)$$

$$RES_k = SBV \cdot \left[1 - \left(1 - \frac{FACT}{LIFE}\right)^{k}\right]$$

$$RDV_k = (SBV - SAL) - RES_k$$

$$RBV_k = RDV_k + SAL = SBV_k - RES_k$$

Where

k = Value for YR

RES_k = Reserve at year k

EXAMPLE: Duplicate the schedule shown. Also calculate the remaining depreciable value in the last year.

NOTE: Note that in the last year of the asset's life there would still be a total of $51,294.43 of remaining depreciable value on the books if this schedule were used throughout the asset's life. (See program 6)

Solution:

```
       40.00 ENT↑
    30000.00 ENT↑
   375000.00 ENT↑
        1.50 GSB1
        1.00 GSB2
             R/S
    14062.50 ***   DEP1
             R/S
   330937.50 ***   RDV1
             R/S
   360937.50 ***   RBV1
             R/S
    14062.50 ***   RES1
             R/S
    13535.16 ***   DEP2
             R/S
   317402.34 ***   RDV2
             R/S
   347402.34 ***   RBV2
             R/S
    27597.66 ***   RES2
       15.00 STO0
             R/S
     8235.18 ***   DEP15
             R/S
   181369.51 ***   RDV15
             R/S
   211369.51 ***   RBV15
             R/S
   163630.49 ***   RES15
       40.00 STO0
             R/S
     3167.32 ***   DEP40
             R/S
    51294.43 ***   RDV40
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter data				
2a.	Life of asset in years	LIFE	ENT↑		
2b.	Salvage value	SAL	ENT↑		
2c.	Starting book value (beginning basis)	SBV	ENT↑		
2d.	Declining rate factor ($1<FAC\leq 2$)	FACT	GSB	1	
3.	Enter year for which depreciation is to be	YR	GSB	2	
	calculated				
4.	RUN:				
4a.	Depreciation		R/S		DEP
4b.	Remaining depreciable value		R/S		RDV
4c.	Remaining book value		R/S		RBV
4d.	Total depreciation to date (Reserve)		R/S		RES
5.	For next years values, go to step 4a				
6.	For another year's values, enter year and	YR	STO	0	
	go to step 4a				

Program Listings

01	*LBL1	
02	STO5	
03	R↓	
04	STO1	
05	R↓	
06	STO2	
07	R↓	
08	STO3	
09	R/S	
10	*LBL2	
11	STO0	
12	R/S	
13	*LBL5	
14	RCL0	
15	RCL3	
16	1	
17	+	
18	X≤Y?	
19	GTO9	Beyond useful life
20	1	
21	RCL5	
22	RCL3	
23	÷	
24	STO6	
25	-	
26	STO7	
27	RCL0	
28	1	
29	-	
30	Y^x	
31	RCL6	
32	x	
33	RCL1	
34	x	
35	STO8	DEP
36	RCL6	
37	RCL1	
38	x	
39	1	
40	RCL7	
41	RCL0	
42	Y^x	
43	-	
44	RCL1	
45	x	RES
46	STO7	
47	RCL1	
48	RCL2	
49	-	Depreciable value
50	X>Y?	
51	GTO8	
52	STO7	Depreciable value
53	-	
54	RCL8	
55	-	
56	CHS	DEP + depreciable value - RES
57	0	
58	X>Y?	
59	GTO6	
60	X⇄Y	
61	R/S	*** DEP
62	*LBL7	
63	RCL1	
64	RCL2	
65	-	
66	RCL7	
67	-	
68	R/S	*** RDV
69	RCL1	
70	RCL7	
71	-	
72	R/S	*** RBV
73	ISZ	Next year
74	RCL7	
75	R/S	*** RES
76	GTO5	
77	*LBL9	
78	0	
79	÷	Error
80	*LBL8	
81	RCL8	
82	R/S	*** DEP
83	GTO7	
84	*LBL6	
85	R/S	*** DEP = 0
86	GTO7	
87	R/S	

*** "Printx" may be inserted to replace "R/S"

REGISTERS

0 YR	1 SBV	2 SAL	3 LIFE	4	5 FACT
6 FACT/LIFE	7 1-FACT/LIFE RES	8 DEP_k	9	.0	.1
.2		.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

CROSSOVER POINT-DECLINING BALANCE TO STRAIGHT LINE

As indicated in the description and example for program 5, the declining balance method of depreciation may not fully depreciate an asset in the asset's lifetime. In these circumstances there is an optimum point in the useful life where a switch from the declining balance method to the straight line method should be made. This is the "crossover point", the first year in which the depreciation by the straight line method is greater than if depreciation were continued using declining balance method. (In accordance with Internal Revenue Service publication 534, the straight line depreciation is determined by dividing the remaining depreciable value by the remaining useful life.)

Given the starting book value (SBV), salvage value (SAL), useful life expectancy (LIFE), and declining balance factor (FACT), this program calculates the last year that the declining balance method should be used, and the remaining life and remaining book value after this "last year" so that a switch to straight line depreciation can be made. Should there be no optimum crossover point, a zero is displayed. This implies that the declining balance method is "best" for the entire depreciable life.

Thus, this program, the declining balance depreciation program (5) and the straight line depreciation program (3) may be used as follows:

A. This program is used to determine the "crossover point" and associated values.

B. Program 5 is entered and a declining balance depreciation schedule is generated for the early years up to and including the year indicated as being the "last year".

Note that since the depreciation programs use the same storage register conventions, only a value for YR need be keyed in for program 5.

C. Finally, program 3 is entered. The remaining book value at the end of the last "declining balance year" is keyed in for starting book value and the remaining life is keyed in for the asset's life. A straight line depreciation schedule may now be generated for the remaining years.

Note that for this portion of the depreciation schedule the value for "total depreciation to date" (reserve) will be in error by an amount equal to the amount depreciated during the declining balance calculations.

As in program 5 the declining balance factor (FACT) should be entered in factor form (1.25, 1.5, 2.0), not as a percent (125, 150, 200).

Equations:

$$SBV\left(1 - \frac{FACT}{LIFE}\right)^{k-1} \cdot \left(\frac{FACT}{LIFE}\right) > \frac{BV_{k-1}}{LIFE + 1 - k}$$

$$BV_k = SBV - SAL - RES_k$$

$$RES_k = SBV\left[1 - \left(1 - \frac{FACT}{LIFE}\right)^k\right]$$

K = the value for YR

The largest integer value for K which maintains the above inequality is the "last year" to use the Declining Balance depreciation method.

Example:

An asset has a starting book value of $375,000, a 40 year life expectancy, and a projected salvage value of $30,000. Using a 1.5 declining balance factor:

1. Determine the crossover point and the associated remaining life and remaining book value.

2. Generate the depreciation data for the declining balance "last year" with program 5 (Normally the user would generate a full schedule beginning with the 1st year).

3. Switching to the straight line method (program 3), generate the depreciation data for the year following the declining balance "last year".

Solution:

1.50	ENT↑	
40.00	ENT↑	
30000.00	ENT↑	
375000.00	GSB1	
18.00	***	last year
	R/S	
22.00	***	remaining life
	R/S	
188471.01	***	RBV

Key in program 5

18.00	GSB2	
	R/S	
7343.03	***	DEP_{18}
	R/S	
158471.01	***	RDV_{18}
	R/S	
188471.01	***	RBV_{18} (Note agreement with RBV above)
	R/S	
186528.99	***	RES_{18}

Key in program 3

22.00	ENT↑	
30000.00	ENT↑	(the first year of straight line depre-
188471.01	ENT↑	
1.00	GSB1	YR ciation)
	R/S	
7203.23	***	DEP_{19}
	R/S	
151267.78	***	RDV_{19}
	R/S	
181267.78	***	RBV_{19}

etc.

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Enter data:				
2a	• Declining rate factor (1<Fact≤2)	FACT	ENT ↑		
2b	• Life of asset in years	LIFE	ENT ↑		
2c	• Salvage value	SAL	ENT ↑		
2d	• Starting book value (beginning basis)	SBV			
3	Calculate last year to use declining		GSB	1	last year
	balance method				
4	Calculate remaining life		R/S		rem. life
5	Calculate remaining book value		R/S		RBV
6	For a new case go to step 2				

Program Listings

```
01 *LBL1
02  STO1
03    R↓
04  STO2
05    R↓
06  STO3
07    R↓
08  STO5
09     1
10  STO0      start with K = 3
11  RCL5
12  RCL3
13     ÷
14  STO7
15     -
16  STO8
17 *LBL0
18   ISZ
19  RCL0      K - 1
20  RCL3
21     1
22     +
23  X≤Y?
24  GTO9      no crossover point
25  RCL8
26  RCL0
27    Y^X
28     1
29     -
30  RCL1
31     x
32  RCL1
33     +
34  RCL2
35     -
36  STO6      BV
37  RCL3
38  RCL0
39     -
40     ÷      right side of
                inequality
41  RCL8
42  RCL0
43    Y^X
44  RCL7
45     x
46  RCL1
47     x      left side of inequality
48  X>Y?
49  GTO0      iterate
50  RCL0
51   R/S      ***last year
52  RCL3
53  RCL0
54     -
55   R/S      ***rem. life
56  RCL6
57  RCL2
58     +
59   R/S      ***RBV
60 *LBL9
61     0
62   R/S      display "0"
```

***"Print x" may be inserted to replace "R/S".

REGISTERS

0 K - 1	1 SBV	2 SAL	3 LIFE	4	5 FACT
6 BV	7 FACT/LIFE	8 1- FACT/LIFE	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

NOMINAL TO EFFECTIVE/EFFECTIVE TO NOMINAL RATE CONVERSION

An annual effective interest rate demonstrates the effect of compounding for a full year of compounding periods at a particular periodic interest rate. The periodic interest rate to be used is determined by dividing the number of compounding periods in a year into the stated annual nominal interest rate. The effect is such that if the nominal rate is held constant, as the number of compounding periods per year is increased ihe annual effective interest rate will increase. The ultimate or upper limit in this process is to have an infinite number of compounding periods in a year, commonly called continuous compounding.

The first part of the program addresses finite compounding, that is quarterly compounding, monthly compounding, etc. Given the number of compounding periods in a year and one of the rates (nominal or effective) the other rate can be calculated. If for example, you require the periodic interest rate for a calculation, given the effective rate, use this program to determine the annual nominal rate first. Dividing the nominal rate by the number of compounding periods in a year will give the required periodic interest rate.

The latter part of the program is for contininuous compounding. Given either rate, the other can be calculated.

The most common and straightforward definition of effective interest rate has been implemented. Occasionally other definitions will be used and the results will not compare exactly with those calculated by these programs. For example, since the maximum annual nominal rate that savings institutions can offer is regulated by law, they may modify the process (also regulated) so that the effective rate is even higher (e.g., for daily compounding, the periodic rate may be divided by 360 and then compounding accomplished for 365 periods). It is important then, when attempting to match results, to understand the process employed.

EQUATIONS:

finite compounding

$$EFF = \left(1 + \frac{NOM}{C}\right)^{C} - 1$$

continuous compounding

$$EFF = (e^{NOM} - 1)$$

EXAMPLES:

1. An investment with monthly cash flows (implying monthly compounding) is said to have an annual effective yield (interest rate) of 21%. What annual (nominal) yield and periodic yield does this represent?

2. A bank offers a savings plan with a 5% annual nominal interest rate. What is the annual effective rate if compounding is continuous?

SOLUTIONS:

```
100.00 ST00
 12.00 ST01
 21.00 ST03
       GSB1
 19.21  ***  Nom(%),annual

  5.00 ST04
       GSB4
  5.13  ***  Eff_cont(%)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter constant and go to either step 3a	100	STO	0	
	for finite compounding or step 3b for				
	continuous compounding				
3a.	Enter				
	- The number of compounding periods/year	c/yr	STO	1	
	and one of the following:				
	- Annual nominal rate	Nom %	STO	2	
	or				
	- Annual effective rate	Eff %	STO	3	
4a.	Calculate the remaining rate				
	- Annual nominal rate		GSB	1	Nom %
	or				
	- Annual effective rate		GSB	2	Eff %
3b.	Enter one of the following:				
	- Annual nominal rate	Nom %	STO	4	
	or				
	- Annual effective rate for continuous	Eff_{cont}%	STO	5	
	compounding				
4b.	Calculate the remaining rate				
	- Annual nominal rate		GSB	3	Nom %
	or				
	- Annual effective rate for continuous		GSB	4	Eff_{cont}%
	compounding				

Program Listings

```
01 *LBL1
02  RCL3
03  RCL0
04   ÷
05   1
06   +
07  RCL1
08  1/X
09   Yˣ
10   1
11   -
12  RCL1
13   x
14  RCL0
15   x
16  R/S         *** Nom
17 *LBL2
18  RCL2
19  RCL1
20  RCL0
21   x
22   ÷
23   1
24   +
25  RCL1
26   Yˣ
27   1
28   -
29  RCL0
30   x
31  R/S         *** Eff
32 *LBL3
33  RCL5
34  RCL0
35   ÷
36   1
37   +
38   LN
39  RCL0
40   x
41  R/S         *** Nom
42 *LBL4
43  RCL4
44  RCL0
45   ÷
46   eˣ
47   1
48   -
49  RCL0
50   x
51  R/S         *** Eff_cont
```

REGISTERS

0 100	1 C/yr	2 Nom	3 Eff	4 Nom	5 Eff_{cont}
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

LEASE VERSUS PURCHASE

This program calculates the present value of the cost of purchasing, CP, the present value of the cost of leasing, CL, and the net difference using the following equations:

$$CP = P + M\frac{(1+i)^n - 1}{i(1+i)^n} - \frac{SV}{(1+i)^n}$$

$$CL = L\frac{(1+i)^n - 1}{i(1+i)^n}$$

Net Difference = CP - CL

where

- P = purchase price
- M = maintenance costs, per period
- i = the opportunity interest rate, per period
- n = the number of periods (useful life)
- SV = salvage value
- L = lease payments

It also calculates the cost of purchasing after leasing for n periods.

$$OP = \frac{P - \text{Credits}}{(1+i)^n} + (L - M)\frac{(1+i)^n - 1}{i(1+i)^n}$$

where Credits = rental credits applied toward purchase

This program is adapted from HP-65 Users' Library program #01093A by Robert Dudugjian.

Example:

Suppose a purchase price of $14,972, maintenance of $15/month, a salvage value at the end of 84 months of $1,000 and lease payments of $325/mo. Suppose further an opportunity rate of interest of .00757543 per month.

Suppose further that the equipment is leased for 12 months and then purchased with $900 rental credits. What is the cost of doing this above the cost of outright purchasing? Suppose it is leased for 24 months with $2000 rental credits.

Solution:

```
14972.00 STO1
  325.00 STO4
0.00757543 STO5
   15.00 STO6
 1000.00 STO7
   84.00 GSB1
15371.15  ***    CP
          R/S
-4771.33  ***    Net (since the answer is
                  less than 0, it implies
  900.00 ENT↑     it is cheaper to pur-
   12.00 GSB2     chase rather than to
                  lease by $4771.33)
 1424.73  ***    OP1

 2000.00 ENT↑
   24.00 GSB2
 2630.41  ***    OP2 Cost of leasing for
                     24 months before
                     purchasing
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Store data:				
2a	Purchase price	P	STO	1	
2b	Lease payments	L	STO	4	
2c	Opportunity rate of interest	i, per period	STO	5	
2d	Maintenance cost	m, per period	STO	6	
2e	Salvage value	SV	STO	7	
3	Enter useful life	n, periods			
4	Calculate net cost of purchasing		GSB	1	CP
5	Calculate net difference		R/S		CP - CL
6	For a new case, go to step 2				
7	(Optional) Enter:				
7a	Rental credits applied to purchase price	Credits	ENT ↑		
7b	Periods of interim lease	m, periods			
8	Calculate cost of leasing for m periods,		GSB	2	OP
	before purchasing				
9	For another case, go to step 7				

```
01 *LBL1
02  GSB0
03  RCL1
04  RCL7
05  RCL2
06     x
07     -
08  RCL6
09  RCL3
10     x
11     +
12   R/S      *** CP
13  RCL4
14  RCL3
15     x
16     -
17   R/S      *** CP - CL
18 *LBL2
19  GSB0
20  RCL4
21  RCL6
22     -
23     x
24   X⇄Y      Credits
25  RCL1
26     -
27  RCL2
28     x
29     -
30  RCL1
31     -
32   R/S      *** OP
33 *LBL0
34  RCL5
35     1
36     +
37   X⇄Y      n
38    Y^X
39   1/X
40  STO2
41  LSTX      (1 + i)^n
42  ENT↑
43     1
44     -
45     x
46  RCL5
47     ÷
48  STO3
49   RTN
```

*** "Print x" may be inserted.

REGISTERS

0	1 P	2 $1/(1+i)^n$	3 $\frac{(1+i)^n-1}{i(1+i)^n}$	4 L	5 i
6 M	7 SV	8		.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

REAL ESTATE
RENTAL INVESTMENT ANALYSIS

Using the equations below, this program solves for monthly rent or cash flow, gross return on investment, and taxable income.

$$\text{Cash flow \%} = \frac{\text{Rent/month} - \text{Cost/month}}{\text{Investment}/12}$$

Gross growth return = Cash flow % +

$$\frac{(\text{P} \times \text{Inflation rate}) + \text{Equity build-up}}{\text{Investment}}$$

$$\underset{\text{(book value)}}{\text{Depreciation/yr}} = \frac{\text{P} - \text{Value of land}}{\text{depreciable life}}$$

$$\approx \frac{.7\ \text{P}}{20}$$

Taxable gain = Actual cash flow - depreciation

Taxable income (shelter if negative) = Taxable gain x tax bracket

This program is adapted from HP-65 Users' Library program #01216A by John Feemster

Example:

A house is for sale for $30,000 with an assumable 6 3/4% FHA Loan paid down to $23,500. Payments of principle, interest, taxes, and insurance are $239.17 per month. The place will rent for $275.00/month. The investor is in the 30% tax bracket. The inflation rate is 7%. Determine the cash flow %, gross growth return, and taxable income from the investment.

Solution:

```
30000.00 STO1
23500.00   -
 6500.00  ***
         STO2
  239.17 STO3
  275.00 STO4
         GSB1
         GSB3
    6.61  ***  Cash flow (%)

23500.00 ENT↑
    6.75 ENT↑
    7.00 GSB4
   26.37  ***  Return (%)
   30.00 R/S
  199.13  ***  Taxable income ($)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Store data:				
2a	Purchase price	P , $	STO	1	
2b	Investment	I , $	STO	2	
2c	Payment/month	Pmt, $	STO	3	
2d	Either Rent/month	Rent, $	STO	4	
	or Cash flow (% of Investment)	Cash F, %	STO	5	
3	Initialize		GSB	1	
4	Calculate either Rent		GSB	2	Rent/mo, $
	or Cash flow		GSB	3	Cash flow,%
5	Calculate gross return on investment				
5a	Loan amount	Loan, $	ENT ↑		
5b	Interest rate	i , %	ENT ↑		
5c	Inflation factor	INF, %	GSB	4	Gross return, %
6	Calculate { Tax Shelter (if negative)				
	Taxable income				
6a	Tax Bracket	TB, %	R/S		$ taxable/yr

Program Listings

01	*LBL1	initialize
02	1	
03	2	
04	ST×3	annual payments
05	EEX	
06	2	
07	STO0	
08	ST÷5	
09	R/S	
10	*LBL2	calculate rent
11	RCL5	
12	RCL2	
13	×	
14	RCL3	
15	+	
16	1	
17	2	
18	÷	
19	STO4	
20	R/S	*** rent/month
21	*LBL3	calculate cash flow
22	RCL4	
23	1	
24	2	
25	×	
26	RCL3	
27	-	
28	RCL2	
29	÷	
30	STO5	
31	RCL0	
32	×	
33	R/S	*** Cash flow
34	*LBL4	calculate gross return on investment
35	RCL0	
36	÷	
37	RCL1	
38	×	growth due to inflation
39	STO6	
40	R↓	
41	×	approx. interest/yr
42	RCL0	
43	÷	
44	CHS	
45	RCL3	
46	+	
47	STO7	
48	ST+6	
49	RCL2	
50	÷	
51	RCL5	
52	+	
53	RCL0	
54	×	
55	R/S	***gross return
56	RCL0	
57	÷	
58	STO8	
59	RCL1	
60	.	approx. bldg. value
61	7	
62	×	
63	2	
64	0	depreciation, 20 year basis
65	÷	
66	CHS	
67	RCL5	
68	RCL2	
69	×	
70	+	
71	RCL7	
72	+	
73	RCL8	
74	×	
75	R/S	*** taxable $

*** "Print x" may be inserted.

REGISTERS

0 100	1 Purch. price	2 Investment	3 Ann. pmts.	4 Rent/mo.	5 Cash flow
6 used	7 Equity gain	8 Tax bracket	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

BREAK-EVEN ANALYSIS

This program solves the following equations for Break-Even point in units (BEP_u), Break-even point in dollars (BEP_D), Margin of Safety Ratio (M), and Profit or Loss:

Computation Based on Units

1) $BEP_u = \frac{F}{S-V}$

2) $M = \frac{u-BEP_u}{u}$

3) Profit or Loss = u(S-V)-F

Computation Based on Dollars

1) $BEP_D = \frac{F}{R}$

2) $M = \frac{D-BEP_D}{D}$

3) Profit or Loss = DR-F

where

F = Total fixed costs

V = Variable cost per unit

S = Sales price per unit

u = Expected sales in units

D = Expected sales in dollars

R = Marginal income ratio = (S-V)/S

NOTE: The margin of safety will generally have no meaning if expected sales are less than sales at the break-even point.

This program is adapted from HP-65 Users' Library program #01275A by Louis Martinez.

EXAMPLE:

The Delux Publishing Company publishes a magazine with variable costs of $0.40 and a sales price of $0.50. The company has annual fixed cost of $1,000,000.

Compute the following:

1) Break-even point in (a) units and (b) dollars.

2) (a) Profit or loss and (b) Margin of safety ratio for expected sales of 12,500,000 magazines.

3) (a) Profit or loss and (b) Margin of safety ratio for expected sales of $20,000,000.

4) Sales volume in (a) units and (b) dollars needed to generate a profit of $5,000,000.

SOLUTION:

```
       0.40 ENT↑
       0.50 ENT↑
 1000000.00 GSB1
10000000.00  ***   BEPu
             R/S
 5000000.00  ***   BEPD
12500000.00 GSB2
  250000.00  ***   Profit
             R/S
       0.20  ***   M
20000000.00 GSB3
 3000000.00  ***   Profit
             R/S
       0.75  ***   M
 5000000.00 ST+1
            GSB4
60000000.00  ***   BEPu
             R/S
30000000.00  ***   BEPD
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter data:				
2a.	Variable cost per unit	V	ENT↑		
2b.	Sales price per unit	S	ENT↑		
2c.	Fixed costs	F			
3.	Compute break-even point in units		GSB	1	BEP_u
4.	Compute break-even point in dollars		R/S		BEP_D
5.	ENTER:				
5a.	Expected sales in units	u			
5b.	Compute profit (or loss if negative)		GSB	2	Profit or Loss
5c.	Compute margin of safety ratio		R/S		M
6a.	Expected sales in dollars	D			
6b.	Compute profit (or loss if negative)		GSB	3	Profit or Loss
6c.	Compute margin of safety ratio		R/S		M
7.	(Optional) To compute sales volume				
	necessary to provide a desired profit:				
7a.	Add the desired profit to fixed costs	DES. Profit	STO	+1	
7b.	Compute break-even point in units		GSB	4	BEP_u
7c.	Go to step 4				
8.	For a new case, go to step 2				

Program Listings

01 *LBL1			
02 STO1			*** M
03 R↓			
04 STO2			
05 X⇄Y			
06 STO3			
07 -	S-V		
08 STO4			
09 RCL2			
10 ÷			
11 STO5	R		
12 *LBL4			
13 RCL1			
14 RCL5			
15 ÷			
16 STO7	BEP_D		
17 RCL1			
18 RCL4			
19 ÷			
20 STO6			
21 R/S	*** BEP_u		
22 X⇄Y			
23 R/S	*** BEP_D		
24 *LBL2			
25 STO8			
26 RCL4			
27 x			
28 RCL1			
29 -			
30 R/S	*** Profit or Loss		
31 RCL8			
32 RCL6			
33 -			
34 RCL8			
35 ÷			
36 R/S	*** M		
37 *LBL3			
38 STO8			
39 RCL5			
40 x		*** "Printx" may be inserted to replace "R/S".	
41 RCL1			
42 -			
43 R/S	*** Profit or Loss		
44 RCL8			
45 RCL7			
46 -			
47 RCL8			

REGISTERS

0	1 F	2 S	3 V	4 S-V	5 R
6 BEP_u	7 BEP_D	8 u or D	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

NOTES

NOTES

NOTES

In the Hewlett-Packard tradition of supporting HP programmable calculators with quality software, the following titles have been carefully selected to offer useful solutions to many of the most often encountered problems in your field of interest. These ready-made programs are provided with convenient instructions that will allow flexibility of use and efficient operation. We hope that these Solutions books will save your valuable time. They provide you with a tool that will multiply the power of your HP-19C or HP-29C many times over in the months or years ahead.

Mathematics Solutions
Statistics Solutions
Financial Solutions
Electrical Engineering Solutions
Surveying Solutions
Games
Navigational Solutions
Civil Engineering Solutions
Mechanical Engineering Solutions
Student Engineering Solutions

HEWLETT hp PACKARD

Reorder No. 00029-14003 Printed in U.S.A. 00029-90007